

Using Leslie matrices as application of eigenvalues and eigenvectors in a first course in Linear Algebra

Michael Monagan

Department of Mathematics, Simon Fraser University
`mmonagan@cecm.sfu.ca`

Abstract. Leslie matrices may be used to model the age distribution of a population as well as population growth. The dominant eigenvalue tells us the long term population growth and the corresponding eigenvector tells us the long term age distribution. Because the model is so simple, and it does not require any knowledge of physics or chemistry or biology, it's ideal for presenting in a first course on Linear Algebra as the main application of eigenvalues and eigenvectors.

In this paper we present the Leslie age distribution model and provide accompanying exercises suitable for students. We use Maple for both numerical calculations and symbolic calculations. We include some data for real populations that instructors may use for classroom presentation or for assignments.

1 Introduction

Linear algebra is my favorite subject to teach. Like most lower division mathematics courses, it is packed with topics that someone wants to be covered. So there's not much room for applications and certainly no room for applications that require additional mathematics to be introduced first. An application needs to fit in one lecture or less. For linear systems, I like to use Markov matrices as the application as they also introduce a family of matrices. What application should we use to illustrate eigenvalues and eigenvectors? I have 19 linear algebra texts on my office shelf. The most common application for eigenvalues and eigenvectors is to solving linear systems of first order differential equations. A problem with this application is that many students will not yet have seen first order linear systems of differential equations. And for those who have, their understanding will likely be superficial. So this is not a good choice.

We need an application where the model is simple to understand and there are interesting questions that the student can easily explore. I propose that we use Leslie matrices and the Leslie age distribution model. This model is popular in ecology and demographics. It takes 10 to 15 minutes to understand the model and see how to express it as a linear transformation. It takes 15 to 20 minutes to compute the dominant eigenvalue and eigenvector for an example and give a physical interpretation of what the dominant eigenvalue and eigenvector mean for the population. That leaves time to pose some interesting questions.

Four of my texts, Anton and Rorres [1], Poole [8], Lay et. al. [5], Boyd and Vandenberghe [2] use Leslie matrices as an application of eigenvalues and eigenvectors. The latter two do so under the label “linear dynamical systems”. Anton and Rorres give a longer theoretical treatment, data from actual populations, and study what happens when we harvest from the population.

In section 2 I develop the Leslie age distribution model, calculate the dominant eigenvalue and eigenvector for an example in Maple and give a physical interpretation of them. In section 3 I explore the Leslie matrix from an algebraic viewpoint using Maple to do some of calculations. In section 4 I explore some questions about controlling the growth of a population. For sections 2–4, I have included some exercises that can be used for student assignments. I have also gathered some data from real populations from the literature in the Appendix.

2 The Leslie Population Distribution Model

Leslie matrices model the age distribution of a population over time. They model births, the aging process and deaths of a human or animal population. I think the best way to introduce this subject is to present an actual example and show how the model leads to a matrix times a vector before presenting the general case. A real example will focus the attention of the students. I use the example of the grey seal population on Sable island, an island off the coast of Nova Scotia. The data for the example is taken from [7]. The model is presented in Figure 1.




Fig. 1. Leslie model for grey seal population

We divide the females of the population into n age groups G_1, G_2, \dots, G_n . In Figure 1 we have divided the seal population into three age groups, G_1 ,

seal pups, ages 0 – 4 years, G_2 young seal adults ages 4 – 8 years and mature seal adults aged over 8 years. We model the fertility rates f_1, f_2, \dots, f_n which are the average number of female births per female in the time period. In our seal population we have $f_1 = 0$ which means seal pups are not mature enough to reproduce, $f_2 = 1.26$ for young adult female seals and $f_3 = 2.0$ for mature female seals which means each seal has on average two female seal pups in 4 years, so one seal pup per year until they die. We also model average survival rates for each age group. In our seal population these are $s_1 = 0.614$, $s_2 = s_3 = 0.808$ meaning over 60% of seal pups survive to be 4 years old. These numbers (see [7]) are estimates based on that has been gathered by scientists over a long period.

In [5] Lay et. al. study the northern spotted owl population. They also divide the owl population into three age groups with fertility rates $f_1 = 0.0$, $f_2 = 0.0$, $f_3 = 0.33$ and survival rates $s_1 = 0.18$, $s_2 = 0.71$, $s_3 = 0.94$.

Let p_i^t be the number of females in age group i at time t . So the population vector at time t is $P^{(t)} = [p_1^t, p_2^t, \dots, p_n^t]$. It is the female population at time t . According to the model the population at time $t + 1$ is given by

$$P^{(t+1)} = \begin{bmatrix} f_1 p_1^t + f_2 p_2^t + f_3 p_3^t \\ s_1 p_1^t \\ s_2 p_2^t + s_3 p_3^t \end{bmatrix}$$

The key observation is that the model is a linear transformation so we may write $P^{(t+1)} = LP^{(t)}$ for some $n \times n$ matrix L , a Leslie matrix. One should spend some time constructing the matrix L here so the student can see where the matrix comes from and why Linear Algebra is involved.

$$\begin{bmatrix} p_1^{(t+1)} \\ p_2^{(t+1)} \\ p_3^{(t+1)} \end{bmatrix} = \begin{bmatrix} f_1 & f_2 & f_3 \\ s_1 & 0 & 0 \\ 0 & s_2 & s_3 \end{bmatrix} \begin{bmatrix} p_1^t \\ p_2^t \\ p_3^t \end{bmatrix}$$

The Leslie matrices for the grey seal population and northern spotted owl population are given below in Figure 2. For the reader, what we will eventually find is that both matrices have a dominant eigenvalue λ^+ . For the grey seals $\lambda^+ = 1.49$ which means the seal population is growing rapidly (it is exploding) and because of this there is an effort to stop the population growing. For the owl population $\lambda^+ = 0.91$ which means the owl population is dying and there is a concerted effort to save it. For teaching we proceed as follows.

$$\begin{bmatrix} 0 & 1.26 & 2.0 \\ 0.614 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix} \begin{bmatrix} 0 & 0 & 0.33 \\ 0.18 & 0 & 0 \\ 0 & 0.71 & 0.94 \end{bmatrix}$$

Fig. 2. Leslie matrices for Sable island grey seals and northern spotted owls

We are interested in the distribution of the population among the n age groups. Let us define population distribution vector $D^{(t)}$ to be $P^{(t)}/p_t$ where $p_t = \sum_{i=1}^n p_i^t$ is the total female population at time t .

Suppose the current seal population is $P^{(0)} = [1, 1, 1]$ thousands. We calculate $P^{(1)}$ using Maple as follows.

```
> L := Matrix([[0.0,1.26,2.00],[0.614,0,0],[0,0.808,0.808]]);
```

$$L := \begin{bmatrix} 0.0 & 1.26 & 2.0 \\ 0.614 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix}$$

```
> P[0] := <1,1,1>: P[1] := L.P[0];
```

$$P_1 := \begin{bmatrix} 3.260000000 \\ 0.6140000000 \\ 1.616000000 \end{bmatrix}$$

After 16 time periods (64 years) so we get the following data.

P^{16}	D^{16}	P^{15}	D^{15}
$\begin{bmatrix} 1115.126895 \\ 459.0148362 \\ 542.5083059 \end{bmatrix}$	$\begin{bmatrix} 0.5268357429 \\ 0.2168591067 \\ 0.2563051504 \end{bmatrix}$	$\begin{bmatrix} 747.5811664 \\ 307.7235768 \\ 363.6975940 \end{bmatrix}$	$\begin{bmatrix} 0.5268357541 \\ 0.2168591050 \\ 0.2563051409 \end{bmatrix}$

Maple code to compute these vectors is

```
> local D; # by default D is the differential operator in Maple
> t := 15;
> pop := proc(v) local i; add(v[i],i=1..numelems(v)) end;
> for i to t do P[t] := L.P[t-1]; D[t] := P[t]/pop(P[t]); od;
```

Now we connect what has happened to the seal population with the dominant eigenvalue and eigenvector of L . First, the seal population has exploded! It has increased from 3 thousand to 2,116 thousand. Second, comparing $D^{(15)}$ and $D^{(16)}$, the population age distribution has stabilized at 52.7% seal pups, 21.7% young adults, and 25.6% mature adults. Consider the quantities

$$\frac{P_1^{16}}{P_1^{15}} = 1.491646586, \quad \frac{P_2^{16}}{P_2^{15}} = 1.491646630, \quad \text{and} \quad \frac{P_3^{16}}{P_3^{15}} = 1.491646673.$$

This means $\lambda = 1.4916466$ and $v = D^{15}$ satisfy $Lv = \lambda v$ to 7 decimal places. Thus the sequence D_1, D_2, D_3, \dots is converging to an eigenvector of L . We state the following Theorem for a Leslie matrix L .

Theorem 1. For any non-zero initial population $P^0 = [p_1^0, p_1^0, \dots, p_n^0]$, if at least one fertility rate f_i is positive, the Leslie matrix L has a unique positive eigenvalue λ^+ . If v^+ is the corresponding eigenvector and at least two consecutive fertility rates are positive, λ^+ is dominant and the population distribution will converge to an eigenvector of L , that is $\lim_{t \rightarrow \infty} D^{(t)}$ exists and is a multiple of v^+ .

We also have the following physical interpretation for λ^+ .

$\lambda^+ < 1$ means the population will decline exponentially.

$\lambda^+ > 1$ means the population will grow exponentially.

$\lambda^+ = 1$ means the population is stable, it does not change.

Below we calculate the eigenvalues of L and the dominant eigenvector of L using Maple. This is too difficult to do by hand. In exercise 3 below, I've constructed L with $\lambda^+ = 7/6$ so that a hand calculation is easy.

```
> with(LinearAlgebra):
> E := Eigenvalues(L);
> lambda := 1.49164663488336:
> I3 := IdentityMatrix(3):
> v := NullSpace(L-lambda*I3)[1]:
> E, v/pop(v);
```

$$\begin{bmatrix} -0.341823317441679 + 0.359549749028222i \\ -0.341823317441679 - 0.359549749028222i \\ 1.49164663488336 + 0.i \end{bmatrix}, \begin{bmatrix} 0.526835747502870 \\ 0.216859101480196 \\ 0.256305151016934 \end{bmatrix}$$

Exercises

- 1 For a population with two age groups with $f_1 = 1$, $f_2 = 1$, $s_1 = 0.75$ and $s_2 = 0$. Write down the Leslie matrix. Calculate the eigenvalues and eigenvectors. Is the population growing or declining? What is the long term population distribution?
- 2 Calculate the positive eigenvalue of L for the spotted owl population. See Figure 2. Is the population growing or dying?
- 3 For the Leslie matrix below calculate the eigenvalues. You should find that one is 0 and one is positive. For the positive eigenvalue, determine the corresponding eigenvector. What is the long term population distribution vector?

$$L = \begin{bmatrix} 0 & 7/6 & 7/6 \\ 1/2 & 0 & 0 \\ 0 & 2/3 & 2/3 \end{bmatrix}$$

- 4 For the northern spotted owl population (see Figure 2), starting with $P^0 = [0.2, 0.1, 0.7]$, calculate $P^4 = L^4 P^0$ and $P^5 = L^5 P^0$ and determine the age distribution. To how many decimal places has the population distribution converged. Estimate the corresponding eigenvalue.
- 5 This exercise is taken from Poole [8]. Woodland caribou are found primarily in western Canada and the American northwest. The fertility rates and survival rates are given in the table below. The data shows that caribou cows do not give birth during their first two years and the survival rate for caribou calves is low.

Age	0-2	2-4	4-6	6-8	8-10	10-12	12-14
f_i	0.0	0.2	0.9	0.9	0.9	0.8	0.3
s_i	0.3	0.7	0.9	0.9	0.9	0.6	0.0
$P_i^{(0)}$	10	2	8	5	12	0	1

Construct the Leslie matrix. Shown also in the last row is the female caribou population in Jasper National park in 1990. Predict the female population in 1992, 1994, 1996, 1998 and 2000. What do you conclude will happen to the population in the long term? Use a computer to compute the eigenvalues of L . What is λ^+ ? What does this tell you about the population?

3 The Leslie Matrix

A Leslie matrix is an n by n matrix of the form

$$L = \begin{bmatrix} f_1 & f_2 & \cdots & f_{n-1} & f_n \\ s_1 & 0 & \cdots & 0 & 0 \\ 0 & s_2 & \cdots & 0 & 0 \\ \vdots & \vdots & & \vdots & \vdots \\ 0 & 0 & \cdots & s_{n-1} & 0 \end{bmatrix} \quad (1)$$

where $n \geq 2$, the survival rates $s_i > 0$ and fertility rates $f_i \geq 0$ with at least one $f_i > 0$. Thus the Fibonacci matrix $[[1, 1], [1, 0]]$ is a Leslie matrix. Notice that we have $s_n = L_{nn} = 0$. If $s_n > 0$, as was the case for the grey seals and spotted owls, we say L is a generalized Leslie matrix. Here we assume $s_n = 0$. One of the exercises in [1] is to show that the characteristic polynomial of L is

$$c(\lambda) = \lambda^n - f_1 \lambda^{n-1} - s_1 f_2 \lambda^{n-2} - s_2 s_1 f_3 \lambda^{n-3} - \cdots - s_{n-1} \cdots s_3 s_2 s_1 f_n \quad (2)$$

The difficulty is that the matrix has an arbitrary dimension. To do this we would suggest that the student first calculate $c(\lambda)$ for $n = 2$ and $n = 3$. Using Maple it is easy to do this. We will do it first for $n = 3$ then for $n = 4$.

```
> with(LinearAlgebra):
> L := Matrix([[f[1],f[2],f[3]], [s[1],0,0], [0,s[2],0]]);
```

$$\begin{bmatrix} f_1 & f_2 & f_3 \\ s_1 & 0 & 0 \\ 0 & s_2 & 0 \end{bmatrix}$$

```
> CharacteristicPolynomial(L,x);
```

$$x^3 - x^2 f_1 - x f_2 s_1 - s_2 s_1 f_3$$

```
> C := CharacteristicMatrix(L,x);
```

$$C := \begin{bmatrix} x - f_1 & -f_2 & -f_3 & -f_4 \\ -s_1 & x & 0 & 0 \\ 0 & -s_2 & x & 0 \\ 0 & 0 & -s_3 & x \end{bmatrix}$$

```
> Determinant(C);
```

$$x^4 - f_1 x^3 - s_1 f_2 x^2 - s_2 s_1 f_3 x - s_3 s_2 s_1 f_4$$

Anton and Rorres [1] give the following formula for the eigenvector of L .

$$v = \begin{bmatrix} 1 & \frac{b_1}{\lambda} & \frac{b_1 b_2}{\lambda^2} & \frac{b_1 b_2 b_3}{\lambda^3} & \dots & \frac{b_1 b_2 b_{n-1}}{\lambda^{n-1}} \end{bmatrix}^T \quad (3)$$

How would we check this? The right way is simply to calculate Lv and λv and try to show $Lv = \lambda v$. Another not so clever way, which I confess to trying at first, is to try to calculate the eigenvector, that is, solve $L - \lambda z = 0$ for z in terms of λ and try to show that $v = sz$ for some scalar s . Again, using Maple, we can only do this for a fixed n . Let us try the first way for $n = 4$.

```
> v := <1,s[1]/x,s[1]*s[2]/x^2,s[1]*s[2]*s[3]/x^3> :
> y := L.v:
> y, x*v;
```

$$\begin{bmatrix} f_1 + \frac{s_1 f_2}{x} + \frac{s_2 s_1 f_3}{x^2} + \frac{s_3 s_2 s_1 f_4}{x^3} \\ s_1 \\ \frac{s_2 s_1}{x} \\ \frac{s_3 s_2 s_1}{x^2} \end{bmatrix}, \begin{bmatrix} x \\ s_1 \\ \frac{s_2 s_1}{x} \\ \frac{s_3 s_2 s_1}{x^2} \end{bmatrix}$$

It seems that all I have to do is check that $y_1 = x$. It is tempting to try to manipulate y_1 to get x . It is better to show try to show $y_1 - x = 0$, that is, to show that $y_1 - x \bmod c(x) = 0$. This is always true if you are using a computer algebra system. It is also often true for a hand calculation. How do we tell Maple to simplify $y_1 - x$ using the constraint $c(x) = 0$? One way to do this is to use the `simplify` command directly as follows

```
> simplify(y[1]-x,{c = 0});
```

0

If the second input to the **simplify** command is a set of algebraic equations, they are treated as constraints. Alternatively one could use division. First multiply $y_1 - x$ by x^3 to clear the denominators so that $x^3(y_1 - x)$ is a polynomial in x then and divide $x^3(y_1 - x)$ by $c(x)$ to get the remainder.

```
> zero := numer(y[1]-x);
```

$$f_1 x^3 + f_2 s_1 x^2 + f_3 s_1 s_2 x + f_4 s_1 s_2 s_3 f_4 - x^4$$

This is just the negative of the characteristic polynomial.

```
> rem(zero,c,x);
```

0

The remainder command treats the inputs as polynomials in x .

Another way is tell Maple that $c(x) = 0$ directly. I will use a Maple **RootOf** to do this. The way to read the following command is that λ one of the roots of $c(x) = 0$ and λ is how this root will be displayed. Then we use Maple's **evala** facility to evaluate algebraic expressions.

```
> alias( lambda=RootOf(c,x) );
```

```
> evala( lambda^4-f[1]*lambda^3 );
```

$$s_1 (f_2 \lambda^2 + f_3 \lambda s_2 + f_4 s_2 s_3)$$

```
> evala( subs(x=lambda,y[1]) );
```

λ

Now for the not so smart way. We've just told Maple that λ is a root of the characteristic polynomial. Let's calculate the eigenvector the way we teach a student to do it by solving $(L - \lambda I)u = 0$ for u in terms of the f_i and s_i . Since the system is homogenous it should have a free parameter. In the Maple code below I tell Maple to use t for the parameter and, to save space, I print u^T .

```
> I4 := IdentityMatrix(4):
```

```
> u := LinearSolve(L-lambda*I4,<0,0,0,0>,free=t):
```

```
> Transpose(u);
```

$$\begin{bmatrix} \frac{\lambda^3 t_4}{s_3 s_2 s_1} & \frac{\lambda^2 t_4}{s_3 s_2} & \frac{\lambda t_4}{s_3} & t_4 \end{bmatrix}$$

Let's try $t = v_4$.

```
> u := subs( t[4]=v[4], u );
```

```
> Transpose(u);
```


$$\begin{bmatrix} 1 & \frac{s_1}{\lambda} & \frac{s_1 s_2}{\lambda^2} & \frac{s_1 s_2 s_3}{\lambda^3} \end{bmatrix}$$

Well that worked with no further simplification required.

To show that the Leslie matrix L has one positive eigenvalue λ_1 we introduce

$$q(\lambda) = \frac{f_1}{\lambda} + \frac{f_2 s_1}{\lambda^2} + \frac{f_3 s_1 s_2}{\lambda^3} + \cdots + \frac{f_n s_1 s_2 \cdots s_{n-1}}{\lambda^n} \quad (4)$$

and claim $q(\lambda) = 1$ for $\lambda \neq 0$ which we leave as an exercise. Now since $f_i \geq 0$ and $s_i > 0$ the function $q(\lambda)$ is monotonically decreasing and $\lim_{\lambda \rightarrow \infty} q(\lambda) = 0$. Consequently there is only one λ , say $\lambda = \lambda^+$ such that $q(\lambda^+) = 1$. That is, L has a unique positive eigenvalue λ^+ . Left as an additional exercise is to show that λ^+ has multiplicity 1. Anton and Rorres go on to investigate when λ^+ is a dominant eigenvalue.

Exercises

- 1 Show that characteristic polynomial for an n by n Leslie matrix given by equation (1) is (2).
- 2 Show that $q(\lambda) = 1$.
- 3 Show that the positive eigenvalue λ_1 of a Leslie matrix has algebraic multiplicity 1. Hint: a root λ_1 of a polynomial $q(\lambda)$ has multiplicity 1 if and only if $q'(\lambda_1) \neq 0$.
- 4 For a generalized Leslie matrix

$$L = \begin{bmatrix} f_1 & f_2 & f_3 \\ s_2 & 0 & 0 \\ 0 & s_2 & s_3 \end{bmatrix}$$

use Maple to calculate the characteristic polynomial $c(x)$. Now try to find a formula for the $c(x)$ for an n by n generalized Leslie matrix.

- 5 The net reproduction rate of a population is defined as

$$r = f_1 + f_2 s_1 + f_3 s_1 s_2 + \cdots + f_n s_1 s_2 \cdots s_{n-1}.$$

Explain why r can be interpreted as the average number of daughters born to a female over her lifetime. It follows that if $r > 1$ the population will grow but if $r < 1$ it will decline. Calculate r for caribou population in Section 2 Exercise 5.

4 Population stabilization and harvesting

Consider the Leslie matrix for the grey seal population.

$$\begin{bmatrix} 0 & 1.26 & 2.0 \\ 0.614 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix}$$

We have determined that the dominant eigenvalue $\lambda^+ = 1.49$ which means the seal population is growing by almost 50% every four years. How can we stabilize the population so that it is neither growing nor declining? The idea is to change the fertility rates f_1, f_2, f_3 or the survival rates s_1, s_2, s_3 to force $\lambda^+ = 1$. We consider two possibilities.

- 1 Reduce s_1 by culling the seal pups every 4 years.
- 2 Reduce all f_i by shooting all seals with infertility darts.

I do the first experiment in class by hand and leave the second as an exercise. Here I will run the both experiments using Maple.

```
> L := Matrix([[0.0,1.26,2.0],[s[1],0,0],[0,0.808,0.808]]);
```

$$L := \begin{bmatrix} 0 & 1.26 & 2.0 \\ s_1 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix}$$

Now force the eigenvalue $x = 1$ and solve for s_1 .

```
> I3 := IdentityMatrix(3):
```

```
> C := L - I;
```

$$C := \begin{bmatrix} -1. & 1.26 & 2.0 \\ s_1 & -1. & 0 \\ 0 & 0.808 & -0.192 \end{bmatrix}$$

```
> c := Determinant(C);
```

$$c := -0.192 + 1.85792 s_1$$

```
> s[1] = solve( c=0, s[1] );
```

$$s_1 = 0.1033413710$$

So we must reduce s_1 from 61% to 10% to stop the population growing. Such a huge reduction indicates how healthy the population is. Continuing with the second option.

```
> L := Matrix([[0,s*1.26,s*2.0],[0.614,0,0],[0,0.808,0.808]]);
```

$$L := \begin{bmatrix} 0 & 1.26 s & 2.0 s \\ 0.614 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix}$$

```
> c := Determinant(L-I3);
```

$$c := -0.192 + 1.14076288 s$$

```
> s = solve(c=0,s);
```

$$s = 0.1683084218$$

Again, a drastic reduction in the fertility rates is needed to stabilize the population.

Another kind of question that one can ask is, what is the maximal sustainable harvest rate h . That is what value of h can we use such that the matrix

$$L = \begin{bmatrix} 0 & 1.26(1-h) & 2.0(1-h) \\ 0.614(1-h) & 0 & 0 \\ 0 & 0.808(1-h) & 0.808(1-h) \end{bmatrix}$$

has an eigenvalue 1? The answer below is one third.

```
> L := Matrix([[0,1.26,2.0],[0.614,0,0],[0,0.808,0.808]]):
> c := Determinant( (1-h)*L - I3 );
```

$$c := 0.94876288 - 3.45664864h + 1.87500864h^2 - 0.36712288h^3$$

```
> h = fsolve( c=0, h );
```

$$h = 0.3295999357$$

Exercises

- 1 For the grey seal population in Figure 2, what is the maximum sustainable harvesting rate assuming we do not harvest seal pups. Why might it be risky to harvest at this rate?
- 2 For the northern spotted owl population in Figure 2, what must s_1 be so that the owl population stabilizes? Comment on the stability of the population.
- 3 If the government tries to eradicate northern owl predators so that all s_1, s_2, s_3 increase, what rate must they increase by to stabilize the population?

References

1. Howard Anton and Chris Rorres. *Elementary Linear Algebra with Applications* Wiley, 1987.
2. Stephen Boyd and Lieven Vanderberghe. *Introduction to Applied Linear Algebra* Cambridge University Press, 2018.
3. G. Caughley. Parameters for Seasonally Breeding Populations. *Ecology* **48**: 834–839, 1967.
4. R.H. Laberson, R. McKelvey, B.R. Noon, and C. Voss. A Dynamic Analysis of the Viability of the Northern Spotted Owl in a Fragmented Forest Environment. *J. Conservation Biology* **6**: 505–512 (1992).
5. David C. Lay, Stephen R. Lay and Judy J. McDonald. *Linear Algebra and its Applications*, 5th ed, Pearson, 2016.
6. Patrick H. Leslie. The use of matrices in certain population mathematics. *Biometrika* **33**(3): 183–212, 1945.

7. Micheline Manske, Carl J. Schwarz and Wayne T. Stobo. The Estimation of the rate of population change of Grey Seals (*Halichoerus grypus*) on Sable Island using a Leslie projection matrix with Capture-Recapture data. Unpublished manuscript.
8. David Poole. *Linear Algebra, a Modern Introduction*. Brooks/Cole, 2003.
9. A.E. York and J.R. Hartley. Pup production following harvest of female northern fur seals. *Canadian Journal of Fisheries and Aquatic Science* **38**, 84–90, 1981.

Appendix

The following data is taken from [3, 1]. It is for a sheep population. The sheep have a lifespan of 12 years so the age groups are 1 year each. The dominant eigenvalue is 1.176.

Age	0–1	1–2	2–3	3–4	4–5	5–6	6–7	7–8	8–9	9–10	10–11	11–12
f_i	.000	.045	.391	.472	.484	.546	.543	.502	.468	.459	.433	.421
s_i	.845	.975	.965	.950	.926	.895	.850	.786	.691	.561	.370	.000

The following data is taken from [1]. The data is for Canadian females in 1965. Because few women over 50 bear children, we ignore those older than 50. The age groups are 5 years each so 10 age groups. The dominant eigenvalue is 1.076.

Age	0–5	5–10	10–15	15–20	20–25	25–30	30–35	35–40	40–45	45–50
f_i	.0000	.00024	.0586	.2861	.4479	.3640	.2226	.1046	.0283	.0024
s_i	.9965	.9982	.9980	.9973	.9969	.9962	.9946	.9918	.9780	–

The following data is taken from [7]. The data is for the grey seal population on Sable island, an island in the Atlantic off Nova Scotia. If one expands the first age group to 4 age groups of 1 year each with $s_i = \sqrt[4]{0.614} = 0.8852$ and $f_i = 0.0$, I get $\lambda^+ = 1.114$. To compress the model into three 4 year age groups, for G_2 (4–8 yrs) use $f_2 = .142 + .948(.347 + .948(.436 + .948(.468))) = 1.26$ and $s_2 = .948^4 = 0.808$.

Age	0–4	4–5	5–6	6–7	7–8	8–9	9–
f_i	0.000	0.142	0.347	0.436	0.468	0.491	0.500
s_i	0.614	0.948	0.948	0.948	0.948	0.948	0.948

The following data is taken from [9]. The data is for northern fur seals. The birth rates include female and male seal pups. I calculate $\lambda^+ = 1.333$.

Age	0–2	2–4	4–6	6–8	8–10	10–12	12–14
f_i	0.00	0.02	0.70	1.53	1.67	1.65	1.56
s_i	0.91	0.88	0.85	0.80	0.74	0.67	0.59
Age	14–16	16–18	18–20	20–22	22–24	24–26	
f_i	1.45	1.22	0.91	0.70	0.22	0.00	
s_i	0.49	0.38	0.27	0.17	0.15	0.00	